

定理証明支援系 Coq を用いたアクチュアリー数学の形式化

SOMPO ひまわり生命保険株式会社

伊藤 洋介¹

概要

定理証明支援系とは、形式的な表現を用いて数学の証明をコンピュータ上で行うツールである。近年はさまざまな定理証明支援系が開発され、現代数学の形式化やプログラムの検証に利用されている。本稿では代表的な定理証明支援系である Coq を紹介するとともに、それを使用したアクチュアリー数学の形式化について報告する。最後に、Coq や本研究の産業界への応用可能性について考察する。

1 定理証明支援系 Coq

20 世紀後半から現代にかけて、数学の証明をコンピュータ上に実装する研究が盛んに行われている。そうしたツールは定理証明支援系と呼ばれ、これまでに数々のソフトウェアが開発されてきた。本節では代表的な定理証明支援系である Coq を紹介するとともに、数学の形式化の意義や現状について論じる。

1.1 定理証明支援系とは

本節では Coq を始めとする定理証明支援系の概要を説明し、それらの代表的な応用例を紹介する。

1.1.1 コンピュータを用いた定理証明

現在の文部科学省の指導要領によれば、日本の義務教育で初めて証明に触れるのは中学校第 2 学年である。平面図形の証明においては、対頂角が等しいことや、平行ならば同位角が等しいこと等の性質を議論の出発点として、論理的な推論を用いて新たな図形の性質を導く。このように、“与えられた命題から、論理的形式に頼って推論を重ね、結論を導き出すこと”を演繹と呼ぶが、平面図形のさまざまな性質を演繹的に導く方法論は、古代ギリシアの数学者 Euclid によって編纂されたいわゆる“ユークリッド原論”によって一応の完成を見、その後長い間西欧諸国における理論体系の手本とされてきた。

図形を研究する幾何学において新しい性質を発見するには、種々の具体例を通じて幾何学的な直感を養うことが重要であろう。一方で、図形の性質を証明する場合には、直感に頼らずあくまで論理的に推論を積み重ねていくことが求められる。もし途中で論理的に飛躍した場合、最終的に誤った結論を導いてしまう恐れがある。

数学における“証明”については 19 世紀頃から活発な研究が開始され、現代では証明自体を数学的対象として研究する“証明論”という数理論理学の一分野がある。証明論においては、英語や日本語といった自然言語で書かれた数学の証明を記号論理の枠組みで形式化する。形式的な証明はごく限られた推論規則によって機械的に行われるため、適切なプログラミングを行うことでコンピュータ上に実装することが可能である。一つ一つの論証のステップをコンピュータが確かめることで、数学における証明に誤りが無いことを保証することができる。このような定理証明の補佐ツールが定理証明支援系である。

数学を形式化する試みについては、数学者によって受け止め方はさまざまである。ラドバウド大学の Freek Wiedijk は AMS (American Mathematical Society) が発刊する雑誌 Notices において次のように述べている [35, pp. 1413–1414; 拙訳].

数学においては今までに 3 度の革命があった:

- 紀元前 4 世紀にギリシア人によって**証明**が導入され, Euclid の“**原論**”により完結されたこと.
- 19 世紀に数学に**厳密性**が導入されたこと. この時期には, 当時厳密性に欠けていた微分積分学が Cauchy 等により厳密化された. また, 同時期には Frege により数理論理学が, Cantor により集合論が発展した.
- 20 世紀後半から 21 世紀初めにかけて, **形式的数学**が導入されたこと.

多くの数学者はこの第 3 の革命が既に起こっていることに気づいておらず, またこの革命が必要であるということにさえ反対するだろう. しかし, 今後数世紀の間に数学者は我々の時代をこの革命の時代として振り返るだろう. その未来においては, 多くの数学者は数学が完全に形式化されない限りそれを信頼の置けるものとは思えないだろう.

通常は紙とペンで行われる数学をコンピュータに実装することに違和感を覚える人は多いかもしれない. しかし, 近年のコンピュータの進展や人工知能の著しい発展を目の当たりにする時, これを数学の推論に応用しよう考えることは, ごく自然なことだと思われる.

1.1.2 定理証明支援系の紹介

定理証明支援系はこれまで多くの種類が開発されており, それぞれ独自の特徴を備えている. ここでは本研究で使用する Coq を中心に解説する. それに加え, その他の代表的な定理証明支援系として Isabelle と Lean を紹介する.

■Coq Coq の基盤となる理論 Calculus of Constructions は 1980 年代に Thierry Coquand と Gérard Huet によって考案された [9]. この理論はその後のさまざまな拡張を経て, 1989 年に Coq の名が冠せられることとなった [32, p. 605]. Coq はその後も長きにわたり世界的な影響を及ぼし続け, 2013 年には世界最大の計算科学系の学会である ACM (Association for Computing Machinery) から ACM ソフトウェアシステム賞を受賞している.

Coq の公式 Web サイト

<https://coq.inria.fr/>

には Coq の概要, インストール方法, リファレンスマニュアル [32], コミュニティへの案内等が掲載されている. また, インストールをせずに Web 上で Coq を体験できる jsCoq² も利用可能である. Coq にはあらかじめ基礎的な数学が形式化されており, 標準ライブラリ³に格納されている. 具体的には

- Arith: 自然数のライブラリ
- ZArith: 整数のライブラリ
- QArith: 有理数のライブラリ
- Reals: 実数・微分積分学のライブラリ
- Sets: 集合論のライブラリ

等がある。これらをインポートすれば、既に形式化された定義や定理についてはそのまま使用することができる。

標準ライブラリに格納されていない理論 (群論等) については別途形式化する必要がある。Coq では今までに多くの貢献者によってさまざまな理論が形式化されており、それらは主に Coq Package Index⁴ から利用可能である。中でも SSReflect/MathComp⁵ は Coq の有用な拡張として広く利用されており、日本においても入門書 [16] が出版されている。本研究では Coq の標準ライブラリに加えて SSReflect/MathComp および Coquelicot⁶ を用いる。

■Isabelle Isabelle は 1986 年に Lawrence C. Paulson により導入された定理証明支援系である [29]。公式 Web サイト

<https://isabelle.in.tum.de/index.html>

からインストール可能であり、リファレンスマニュアルや学習資料等も閲覧可能である。Archive of Formal Proofs⁷ には Isabelle により形式化された定理が多数収録されている。なお、Isabelle は推論の基礎をなす公理系を選べる設計になっており、中でも高階述語論理を基礎とする Isabelle/HOL は現代数学を広範囲にわたり形式化したライブラリを備えている。本稿で紹介している定理証明支援系の中では確率論の形式化が最も進んでおり、2017 年には Jeremy Avigad 等により中心極限定理が Isabelle により形式化されている [5]。

■Lean Lean は 2013 年にマイクロソフトリサーチの Leonardo de Moura によってプロジェクトが開始された新しい定理証明支援系である [10]。公式 Web サイト

<https://leanprover.github.io/>

からインストール可能であるが、オンラインで実行できる JavaScript 版も用意されている⁸。インペリアル・カレッジ・ロンドンの Kevin Buzzard が現代数学の形式化に積極的に関与しており、AMS が発刊する雑誌 Notices において Lean を紹介している [8]。Lean のライブラリである `mathlib`⁹ には専門的な数学も数多く投稿されており [25]、コミュニティも急速に拡大している。成長著しい言語であるが、発展段階ということもあり後方互換性は必ずしも保証されない。


1.1.3 定理証明支援系の応用例

定理証明支援系は今までに多くの応用が試みられ、数々の実績を残してきた。ここではその一部を紹介する。

■四色定理 Coq が重要な役割を果たした数学の難問として四色問題が挙げられる。現在では既に証明が得られているため四色定理と呼ばれるが、その主張は次の通りである。

定理 (四色定理 [4])。平面上の地図は 4 色で塗り分けられる。

詳細に述べると、“いかなる地図も、隣接する領域が異なる色になるように塗るには 4 色あれば十分である”ということである。



四色定理は 1976 年に Kenneth Appel と Wolfgang Haken によって証明が与えられたが、それはコンピュータプログラムを用いて膨大な場合分けを行うものであったため物議を醸した。その後、より改良された証明が発見されたものの、それらは依然としてコンピュータに依存したものであり、プログラム自体にバグがないことを検証するのが困難であった。2000 年になり Georges Gonthier は四色定理の Coq による形式化を開始し、約 5 年もの歳月を経て完成に至った [14]。これは四色定理の主張と証明を全て Coq で記述したということであり、プログラムの正当性は疑いの余地がなくなった。この応用例は、人間による査読が困難な証明の検証にコンピュータが役立つことを示す重要な事例であると言える。

■seL4 定理証明支援系は数学の定理証明だけでなく、実用的なシステムプログラムの品質確保にも応用されている。具体的には、システムが所望の仕様を満たすことを定理証明支援系により証明するのである。中でも seL4 は定理証明支援系による形式的検証を経た世界初の OS (オペレーティング・システム) カーネルとして知られ [23]、高い信頼性を獲得している。なお、この形式的検証には Isabelle が利用された。

1.2 数学の形式化

本節では、数学を形式化するとはどのようなことなのか、具体例を交えながら説明するとともに、数学を形式化することの意義について論じる。また、現代数学の形式化が現時点でどの程度進んでいるのかについても触れる。

1.2.1 形式的証明とは

数学の論理については 19 世紀末から 20 世紀初頭にかけて活発な研究がなされ、現代数学は集合論に基づく記号論理で記述できるという認識が広まっていった。記号論理とは、論理学を形式的な記号で記述する方法のことである。

具体的な例を命題論理で見てみよう。表 1 は現代の論理学で一般的に用いられる記法である。命題¹⁰や論理記号を組合せて得られる文を論理式と呼ぶ。例えば、自営業を営んでいる太郎という人物がいたとして、彼の

表 1 命題論理の記号

論理記号	意味
$\neg P$	P でない
$P \wedge Q$	P かつ Q
$P \vee Q$	P または Q
$P \rightarrow Q$	P ならば Q
$P \leftrightarrow Q$	P の場合かつこの場合に限り Q

iDeCo への加入資格を考えてみる。論文執筆時点において、第 1 号被保険者の iDeCo への加入資格は以下のようになっている¹¹。

- 満 20 歳以上 60 歳未満である。
- 国民年金保険料を納付している。
- 農業者年金基金に加入していない。

今、命題 P_1, P_2, Q, R を次で定める。

P_1 : 太郎は満 20 歳以上である。

P_2 : 太郎は満 60 歳未満である。

Q : 太郎は国民年金保険料を納付している。

R : 太郎は農業者年金基金に加入している。

すると、太郎の加入資格は次のように表せる。

$$(P_1 \wedge P_2) \wedge Q \wedge (\neg R)$$

このような単純な例では記号化の恩恵が分かりにくいだが、さらに複雑な条件になった場合、論理式は主張を簡潔かつ正確に表現する手段として非常に有効である。

また、記号化することで論理式に対する機械的な操作が可能になるという利点もある。例えば、太郎に加入資格がないとはどういうことか、具体的に見てみよう。それには先ほどの論理式を否定すればよい。

$$\neg((P_1 \wedge P_2) \wedge Q \wedge (\neg R))$$

ここで論理学における De Morgan の法則

$$\neg(A \wedge B) \iff (\neg A) \vee (\neg B)$$

および二重否定律¹²

$$\neg(\neg C) \iff C$$

を用いれば、

$$\begin{aligned} \neg((P_1 \wedge P_2) \wedge Q \wedge (\neg R)) &\iff (\neg(P_1 \wedge P_2)) \vee (\neg Q) \vee (\neg(\neg R)) \\ &\iff ((\neg P_1) \vee (\neg P_2)) \vee (\neg Q) \vee R \end{aligned}$$

と変形することができる。すなわち、太郎に iDeCo の加入資格がないとは次のいずれかを満たすということである。

- 太郎は満 20 歳未満である。
- 太郎は満 60 歳以上である。
- 太郎は国民年金保険料を納付していない。
- 太郎は農業者年金基金に加入している。

以上が命題論理の例であるが、現代数学を記述するには命題論理だけでは不十分であり、それを拡張した述語論理の枠組みが必要である。そこで、表 2 に掲載した新たな記号 \forall と \exists を導入する。

表 2 述語論理の記号

論理記号	意味
$\forall xL(x)$	任意の x に対し $L(x)$
$\exists xL(x)$	ある x に対し $L(x)$

例えば先ほどの命題

$$P_1: \text{太郎は満 20 歳以上である.}$$

を主語“太郎”と述語“満 20 歳以上である”に分解することを考えよう. 今,

$$L_1(x): x \text{ は満 20 歳以上である.}$$

と定めれば, 命題 P_1 は $L_1(\text{太郎})$ と表すことができる. このように述語部分を抽出すると, 具体的な対象“太郎”に捉われずに次のような主張を展開することが可能になる.

$$\exists xL_1(x): \text{ある } x \text{ に対し, } x \text{ は満 20 歳以上である.}$$

これはつまり“少なくとも 1 人は満 20 歳以上の人が存在する.”ということの意味する.

以上の枠組みは現代数学を記述する基礎的な言語として使用できる. 例えば“4 の倍数は偶数である.”という命題を表すには, 例えば二項関係記号 $x \mid y$ を“ x は y を割り切る.”と定めた上で

$$\forall x(4 \mid x \rightarrow 2 \mid x)$$

とすればよい.

それでは, このように形式化された命題を“証明”するにはどうすればよいだろうか. 記号論理学においては, 理論の前提となる“公理”と証明を進めるための“推論規則”が規定されており, それらを用いて命題を導くことを“証明”と呼ぶ. 例えば, 論理記号 \wedge に関する推論規則は次の 2 つである.

\wedge 導入則: 命題 P と命題 Q の両方が得られている時, 命題 $P \wedge Q$ を得る.

\wedge 除去則: 命題 $P \wedge Q$ が得られている時, 命題 P と命題 Q を得る.

また, 論理記号 \rightarrow に関する推論規則は次の 2 つである.

\rightarrow 導入則: 命題 P を仮定して命題 Q が得られた時, 命題 $P \rightarrow Q$ を得る.

\rightarrow 除去則: 命題 $P \rightarrow Q$ と命題 P の両方が得られている時, 命題 Q を得る.

これらを用いると, 三段論法

$$((P \rightarrow Q) \wedge (Q \rightarrow R)) \rightarrow (P \rightarrow R)$$

は次のようにして証明できる.

- 1: $(P \rightarrow Q) \wedge (Q \rightarrow R)$ を仮定する.
- 2: P を仮定する.
- 3: 1 行目より $P \rightarrow Q$ と $Q \rightarrow R$ を得る. (\wedge 除去)
- 4: 2 行目と, 3 行目の $P \rightarrow Q$ から Q を得る. (\rightarrow 除去)
- 5: 4 行目と, 3 行目の $Q \rightarrow R$ から R を得る. (\rightarrow 除去)

- 6: 2行目と5行目から, $P \rightarrow R$ を得る. (\rightarrow 導入)
7: 1行目と6行目から, $((P \rightarrow Q) \wedge (Q \rightarrow R)) \rightarrow (P \rightarrow R)$ を得る. (\rightarrow 導入)

この例からも分かるように, 形式化された数学においては許された推論規則以外証明に用いてはならない. 記号論理に初めて触れる者には不便だと感じられるかもしれないが, 推論規則を限定することで論理の飛躍を防止することができ, またコンピュータへの実装も容易になるのである.

なお, 議論の出発点となる公理として何を採用するかは理論体系により異なる. 現代数学は集合論を基礎に展開できると考えられているが, その際に仮定する公理としては Zermelo-Fraenkel の公理系が有名である. ただし, Coq で数学を基礎づける場合, 一般には Zermelo-Fraenkel 集合論と異なる枠組みを用いる. 数学の基礎づけについては専門的な内容となるためこれ以上は立ち入らない. 前述の通り, Coq の標準ライブラリには既に初等的な数学が形式化されているため, 定理証明支援系が基礎としている公理については詳細を把握していなくても応用上問題ないと思われる.

1.2.2 形式化の意義

前節で形式的証明について概説したが, ここで定理証明支援系を用いた形式化の意義について改めて考えてみようと思う.

■無謬性の保証 形式化された数学においては, 前述の通り限られた推論規則でのみ証明を進めることができる. これを定理証明支援系で実装することができれば, その証明には一切の論理的飛躍のないことが確かめられる.

読者の中には, 数学はそもそも厳密な学問なのだから今更それを確認しても無意味だと思われるかもしれない. しかし, 最先端の数学研究においては査読を経た論文に誤りや論理の飛躍が見つかることは実際にある. 例えば Sébastien Gouëzel と Vladimir Shchur は学術誌 Journal of Functional Analysis において, 2013 年に証明された定理を Isabelle で形式化する際に論理の飛躍を発見したことを報告している [15]. 近年の数学は高度に細分化されていることに加え, 他の研究者により証明された結果に依存して新たな定理を証明することも多い. このことは, 新たな論文の査読において誤りを見つけること自体が困難になっていることを意味する. もし定理証明支援系が発展して査読に利用されるようになれば, 誤りを検出する精度が飛躍的に向上することになる. そもそも著者自身が論文を投稿する際に定理証明支援系による形式化を済ませていれば, それは査読を経るよりも遥かに高い信頼性を獲得することになるだろう.

■システムの形式的検証 定理証明支援系はソフトウェアの品質確保にも利用可能である. 例えば, コンピュータプログラムが満たすべき性質を形式論理で記述し, ソースコードがその仕様を満たすことを定理証明支援系で証明する, といった方法がある.

形式的に検証されたソフトウェアの信頼度は非常に高い. 仮に数多くのサンプルテストをクリアしたソフトウェアであったとしても, 想定される全てのパターンを検証しない限りバグの存在は否定できない. しかし, 定理証明支援系を用いると仕様通りにプログラムが動作することを数学的に証明できるため, 形式化された部分に関してはバグのないことが保証される.

実際, 形式的な証明は情報セキュリティで最も高いレベルの国際認証を得る際に必要とされている. “コモンクライテリア” とは情報システム等のセキュリティを評価するための基準を定めた国際規格であるが, そこで定める保証評価レベル (EAL: Evaluation Assurance Level) は EAL1 から EAL7 までの 7 段階に分かれており, 最も高いレベルの EAL7 の認証を取得するには形式的検証を経ることが求められている.

EAL7 (形式的検証済み設計及びテスト)

リスクが非常に高いか、高い資産価値により、さらに高い開発コストが正当化される場合に適用される。EAL7は、EAL6の保証に加え、数学的検証を伴う形式的表現と対応、広範囲のテストを使用する包括的分析を要求する。[19]

ビッグデータや人工知能の発展が著しい現代、ソフトウェアの誤作動が社会に与える影響はますます大きくなってきている。定理証明支援系を用いた形式的検証は一切のバグを除去する手段として今後も期待が高まっていくものと考えられる。

■**自動推論** 定理証明支援に隣接した分野として自動定理証明がある。Coqを始めとした定理証明支援系では、形式的証明を行う際には原則としてユーザが証明のプロセスを指示する。それに対し、自動定理証明機は目的の定理を与えるだけでコンピュータが自動的に推論を行い、形式的証明を完成させる。

自動定理証明も世界的によく研究されており、中でも William McCune による自動定理証明機 EQP を用いた Robbins 予想¹³の解決は、コンピュータが人間の助けを借りずに数学の問題を解いた成功例として知られる[26]。ただ、数学全般で見れば、現時点においてコンピュータは数学者に取って代わるような水準には達していない。現在、自動推論の質やスピードを向上させる研究が行われているが、その方法の一つとして定理証明支援系で形式化された証明を自動定理証明機に学習させる方法がある。Cezary Kaliszyk と Josef Urban は Kepler 予想¹⁴を形式化する Flyspeck プロジェクトによって得られた大量の定義や補題からなるライブラリを、自動定理証明への機械学習に利用している[21]。この例からも分かるように、形式化された数学は人工知能の学習に活用できる貴重な財産なのである。

1.2.3 数学の形式化の進捗

現代数学の形式化はさまざまな定理証明支援系で行われている。表3は Thomas C. Hales が AMS の雑誌 Notices で紹介した主要な定理の形式化に関する一覧である。この記事が書かれたのは 2008 年であり、現在で

表3 形式化された証明の例 [17, p. 1372; 拙訳]

年	定理	定理証明支援系	実装者
1986	第一不完全性定理	Boyer-Moore	Shankar
1990	平方剰余の相互法則	Boyer-Moore	Russinoff
1996	微分積分学の基本定理	HOL Light	Harrison
2000	代数学の基本定理	Mizar	Milewski
2000	代数学の基本定理	Coq	Geuvers et al.
2004	四色定理	Coq	Gonthier
2004	素数定理	Isabelle	Avigad et al.
2005	Jordan 曲線定理	HOL Light	Hales
2005	Brouwer の不動点定理	HOL Light	Harrison
2006	Flyspeck I	Isabelle	Bauer-Nipkow
2007	留数定理	HOL Light	Harrison
2008	素数定理	HOL Light	Harrison

はさらに多くの定理が形式化されている。直近の進捗を把握する目安として、Freek Wiedijk が自身のホームページに掲載している Formalizing 100 Theorems がある¹⁵。ここには数学の主要な 100 の定理が掲載され、現時点でどの程度形式化されているかがまとめられている。論文執筆時点での進捗率は 97% となっており、多くの定理が形式化されていることが分かる。数学の形式化は現役の数学者だけでなく学生を含めた世界中の貢献者によって進められており、古典的な結果だけでなく、最先端の数学についても形式化が意欲的に試みられている。

しかしながら、アクチュアリー数学については今まで形式化が行われていない。本研究は発展の著しい定理証明を生保数理に应用する初めての試みである。

2 アクチュアリー数学の形式化

本研究では定理証明支援系 Coq を用いて生保数理の基本的な定義や公式を形式化し、その内容を **Actuary** パッケージとしてリリースした。本節では Coq の文法を簡単に紹介した上で、**Actuary** パッケージの大まかな内容や応用例について解説する。

2.1 Coq の文法

Coq を用いて数学の定理を証明するには、まず証明したい命題を論理式で記述し、タクティクと呼ばれる命令で証明を組上げていくことになる。Coq は対話型の定理証明支援系であり、証明を完成させるにはステップごとにユーザから指示を送ることになる。通常のプログラミング言語と同じく、Coq を正しく動作させるには決められた文法に従ってスクリプトを記述する必要がある。

まず第一に、Coq で命題を記述するには第 1.2.1 節で紹介した論理式を Coq における記法 (表 4) に従って入力することになる。前述の三段論法

表 4 論理式の記法

論理記号	Coq の記法
\neg	<code>~</code>
\wedge	<code>/\</code>
\vee	<code>\ </code>
\rightarrow	<code>-></code>
\leftrightarrow	<code><-></code>
\forall	<code>forall</code>
\exists	<code>exists</code>

$$((P \rightarrow Q) \wedge (Q \rightarrow R)) \rightarrow (P \rightarrow R)$$

を Coq で表記すれば、次のようになる。

$$((P \rightarrow Q) /\ (Q \rightarrow R)) \rightarrow (P \rightarrow R)$$

続いて三段論法の証明を Coq で形式化することを考える。結論から述べると、次の四角内部がその証明の一例である。

```

1 Theorem syllogism : forall (P Q R : Prop), ((P -> Q) /\ (Q -> R)) -> (P -> R).
2 Proof.
3   intros P Q R.
4   intro H_PQR.
5   intro H_P.
6   destruct H_PQR as [H_PQ H_QR].
7   apply H_QR.
8   apply H_PQ.
9   assumption.
10  Qed.

```

Coq で三段論法を証明する場合は上記のスクリプトを 1 行ずつ実行していけばよい。無事に最後までエラーがなく実行できれば証明が完了したことを意味する。もし途中で推論に誤りがあったり論理の飛躍があったりすると、Coq はエラーを出力するようになっている。

まず 1 行目で証明したい定理をコマンド `Theorem` の後に続けて宣言している。ここで `syllogism` と書かれている部分は定理の名前であり、原則として自由に設定できる¹⁶。コロンの後に証明したい論理式を記述する。なお、`forall (P Q R : Prop)`、の部分は“任意の命題 P, Q, R に対して”と読む¹⁷。Coq は定理を受け取ると証明モードに入り、以下のような出力をする。

```

1 subgoal

=====
forall P Q R : Prop, (P -> Q) /\ (Q -> R) -> P -> R

```

出力画面の二重線はゴールラインと呼ばれる。ゴールラインより下には今証明すべき命題が表示され、これはサブゴールと呼ばれる。次に 3 行目 `intros P Q R.` を入力すると出力画面が以下のように変わる。

```

1 subgoal

P, Q, R : Prop
=====
(P -> Q) /\ (Q -> R) -> P -> R

```

ゴールラインより上には証明に使用できる前提や変数が表示され、これらはコンテキストと呼ばれる。今、命題 P, Q, R がサブゴールからコンテキストに移されたが、この操作は“任意の命題 P, Q, R を取って固定する”ものだと解釈できる。次の 4 行目 `intro H_PQR.` を実行すると、出力画面は以下ようになる。

```

1 subgoal

P, Q, R : Prop
H_PQR : (P -> Q) /\ (Q -> R)
=====
P -> R

```

この命令はサブゴールの十分条件をコンテキストに移す命令である¹⁸。つまり “ $(P \rightarrow Q) \wedge (Q \rightarrow R)$ を仮定

する”ことを意味する。続く `intro H_P`. も同様であり、出力画面は次の通りである。

```
1 subgoal

P, Q, R : Prop
H_PQR : (P -> Q) /\ (Q -> R)
H_P : P
=====
R
```

さて、今求められているのは “ $(P \rightarrow Q) \wedge (Q \rightarrow R)$ と P が成り立つことを前提とした上で R を示す” ということである。仮定 `H_PQR` はそのままでは使えないので \wedge の除去を行う。その命令が 6 行目の `destruct H_PQR as [H_PQ H_QR]`. である。これは仮定 `H_PQR` を分解して左側を `H_PQ` と、右側を `H_QR` と名づける操作である。

```
1 subgoal

P, Q, R : Prop
H_PQ : P -> Q
H_QR : Q -> R
H_P : P
=====
R
```

続く 7 行目の `apply H_QR`. を実行すると、出力画面は次のようになる。

```
1 subgoal

P, Q, R : Prop
H_PQ : P -> Q
H_QR : Q -> R
H_P : P
=====
Q
```

サブゴールが R から Q に変わっている。この操作を解釈するならば、“今仮定 $Q \rightarrow R$ があるので、 R を示すには Q を示せばよい” ということになる。Coq を始めとする定理証明支援系では、このようにサブゴールを変形して証明を進める方式を採用している。我々が数学の証明を書く時は仮定から順番に演繹を行うのが普通なので、結論を変形していく操作には違和感を覚えるかもしれないが、慣れれば特に難しいものではない。次の 8 行目 `apply H_PQ`. も同様であり、サブゴールが Q から P に変わる。

```
1 subgoal

P, Q, R : Prop
H_PQ : P -> Q
H_QR : Q -> R
H_P : P
```

```
=====
P
```

ここまで来ればサブゴール P はコンテキストに含まれる P そのものである。コンテキストから直ちにサブゴールが得られる場合、9 行目のように `assumption.` と入力すればよい。すると出力結果は以下のようになる。

```
No more subgoals.
```

つまり“もう証明すべきサブゴールは残っていない”ということであり、あとは証明が完了したことを示す `Qed.` を入力すれば証明モードを抜ける。このようにして証明された定理は Coq に登録され、以後は定理 `syllogism` を証明の中で使用できるようになる。なお、Coq へ入力した `intro`, `destruct` 等の命令はタクティクと呼ばれる。直訳すれば“戦術”となるが、サブゴールを解消するための戦術という意味でこのように呼ばれているものと思われる。

通常の数学での証明と同じく、証明の方法は一通りではない。今の例では例示のために証明のステップを意図的に細かく記載したが、この程度の推論であれば Coq は自動的に証明を組上げることができる。

```
1 Theorem syllogism' : forall (P Q R : Prop), ((P -> Q) /\ (Q -> R)) -> (P -> R).
2 Proof.
3   intuition.
4   Qed.
```

このように Coq には若干ではあるが自動証明機能が実装されている。自動証明が可能な条件はあるものの、適切なタイミングで使えば形式化の労力を削減する強力な手段となる。

2.2 Actuary パッケージの概要

前節で紹介したような形式的証明を行って得られた補題・定理群をライブラリまたはパッケージと呼ぶ。筆者は Coq を用いて生保数理の記号や基本的な公式を形式化し、一連の結果を Actuary パッケージとして GitHub¹⁹ に公開した。

<https://github.com/Yosuke-Ito-345/Actuary>

本パッケージの Version 2.4 のファイル構成は表 5 のようになっている²⁰。また、Actuary パッケージは Coq の OPAM (OCaml Package Manager) リポジトリにも登録されているので、OPAM をインストールしていれば以下のコマンドで簡単にインストールできる。

```
opam repo add coq-released https://coq.inria.fr/opam/released
opam install coq-actuary
```

形式化する生保数理の内容は日本のアクチュアリー資格試験の教科書 [12, 13] やアメリカの SOA (Society of Actuaries) 資格試験の教科書 [20] および参考書 [24] を始め、インターネット上に公開されている講義ノート [22, 30] を参考にした。

■**アクチュアリー記号** 生保数理で用いられるアクチュアリー記号も Coq に実装している。アクチュアリー記号は補助記号が多く振られる複雑な記法であり、それをテキスト形式で直感的に分かり易く表現するのはなかなか難しいが、試行錯誤を経て表 6 のような形に落ち着いた。

■**生命表の定義** 生命表 l_x は生保数理における種々の計算の基礎であるが、数学的には実数値関数と考えるのが自然である。Coq へ実装する際は、一般的な生存関数 (survival function) の定義も参考にして次のように定式化した。

定義. 実数上の実数値関数 l が生命関数であるとは、次の 4 条件を全て満たすことを言う ²¹:


1. $l_0 > 0$,
2. 任意の負の実数 u に対し $l_u = l_0$ が成り立つ,
3. $\lim_{u \rightarrow \infty} l_u = 0$,
4. l は広義単調減少である。

表 5 Actuary パッケージの概要

ファイル名	内容
Basics.v	生保数理の形式化に使用する基本的な数学の補題を集めたもの
Interest.v	確定年金現価等の金利に関する記号や公式をまとめたもの
LifeTable.v	生命表の定義や基本的な性質をまとめたもの
Premium.v	生命年金現価や保険料に関する記号や計算式および公式をまとめたもの
Reserve.v	責任準備金の記号や公式をまとめたもの
all_Actuary.v	上記のライブラリを全てまとめたもの
Examples.v	Actuary パッケージの応用例を紹介したもの

表 6 アクチュアリー記号 (抜粋)

意味	アクチュアリー記号	Coq の記法
生存数	l_x	<code>\l_x</code>
死亡数	d_x	<code>\d_x</code>
生存確率	${}_tp_x$	<code>\p_{t & x}</code>
死亡確率	${}_tq_x$	<code>\q_{f t & x}</code>
死力	μ_x	<code>\mu_x</code>
定期保険の現価	$A_{x:\overline{n}}^1$	<code>\A_{x'1:n}</code>
生命年金現価	$\ddot{a}_{x:\overline{n}}$	<code>\a''_{x:n}</code>
定期保険の平準純保険料	$P_{x:\overline{n}}^1$	<code>\P_{x'1:n}</code>
定期保険の責任準備金	${}_tV_{x:\overline{n}}^1$	<code>\V_{t & x'1:n}</code>



生命関数は 0 以上の実数に対して定義されればよいが, Coq では実数全体で関数を定義する方が扱い易いため, 便宜上負値も定義域に含めている. 実際の Actuary パッケージにおける生命関数の定義は次の通りである.

```

1  (* life table *)
2  Record life : Type := Life {
3    l_fun :> R -> R;
4    l_0_pos : 0 < l_fun 0;
5    l_neg_nil : forall u:R, u <= 0 -> l_fun u = l_fun 0;
6    l_infty_0 : is_lim l_fun p_infty 0;
7    l_decr : decreasing l_fun
8  }.

```

文法の詳細な説明は省略するが, ここに現れる **Record** は一般のプログラミング言語で言うレコード型と同じ意味合いである. スクリプト中の R は実数型を表しており, 3 行目の $R \rightarrow R$ は実数から実数への関数を意味する. 4 行目から 7 行目は前述の定義に示した 4 条件を形式化したものである.

なお, この定義は最終年齢が ∞ になる場合も含んでいるし, 微分可能性も仮定していない. その意味で非常に広い定義であるが, 生命年金現価等を定義する上では有限和の方が扱い易いので次の性質を仮定する.

- 定義.** 1. 生命関数 l が有限であるとは, $l_x = 0$ となる自然数 x が存在することを言う.
 2. 有限な生命関数 l に対し, $l_x = 0$ となる最小の自然数 x を l の最終年齢と呼ぶ.

これは Actuary パッケージでは次のように形式化されている.

```

1  Definition ages_dead (l:life) : Ensemble nat := fun x:nat => \l[l]_x = 0.
2  Definition l_finite (l:life) := exists x:nat, (ages_dead l x).
3  Definition ult_age (l:life) (l_fin : l_finite l) := sval (leastN l_fin).

```

ここで nat は自然数を意味し, Ensemble は集合を意味する. 上記の定義は次のような意味合いである.

- 1 行目: 生命関数 l に対し, 集合 $\text{ages_dead}(l)$ を $\{x \in \mathbb{N} \mid l_x = 0\}$ で定める.
 2 行目: 生命関数 l に対し, 命題 “ $x \in \text{ages_dead}(l)$ となる自然数 x が存在する.” を $\text{l_finite}(l)$ と書く. (有限性の定義)
 3 行目: 有限な生命関数 l に対し, 集合 $\text{ages_dead}(l)$ の最小値を $\text{ult_age}(l)$ と書く. (最終年齢の定義)

また, 死力の定義等では微分が用いられるため, 必要に応じて次の性質を仮定する.


定義. 生命関数 l が連続微分可能であるとは, 関数 l が実数全体で連続微分可能であることを言う.

```

1  Hypothesis l_C1 : forall u:R, ex_derive l u /\ continuous (Derive l) u.

```

なお、生命関数を実数上の関数とみなすのであれば、最終年齢 ω の手前で生命関数の値が 0 になることも当然考えられる。つまり、 $l_u = 0$ となるための実数 u に関する必要十分条件が $u \geq \omega$ とはならないのである。



このままでは分割払に関する議論を行う際に分母が 0 になる可能性を排除するのが困難である。そこで、今回の形式化にあたり新たに次の記号を導入することとした。

定義. 生命関数 l に対し、 $\psi \in \mathbb{R} \cup \{+\infty\}$ を $\psi := \inf\{u \in \mathbb{R} \mid l_u = 0\}$ で定める。

すると次の補題が得られる。

補題. 1. 生命関数 l が有限ならば $\psi \leq \omega$ が成り立つ。
2. 生命関数 l が連続である時、実数 u について $l_u = 0$ と $\psi \leq u$ は同値である。

以上のように生命関数が明確に形式化できれば、あとは通常の生保数理の方法で死亡数や生存確率等が定まり、それらを用いて生命年金現価や保険料等を定義することができる。

■諸公式の形式化 よく知られているように保険料や責任準備金等の間にはさまざまな関係が成り立っている。代表的な公式としては次のようなものが形式化されている。

$$A_{x:\overline{n}}^{(m)} = 1 - d^{(m)} \ddot{a}_{x:\overline{n}}^{(m)}$$

```
1 Lemma ins_endow_ann_due : forall (m x n : nat), 0 < m -> x < \omega ->
2 \A^{m}_{x:n} = 1 - \d^{m} * \a''^{m}_{x:n}.
```

次はいわゆる Fackler の再帰式である。

$${}_{t-1}V_{x:\overline{n}}^1 + P_{x:\overline{n}}^1 - vq_{x+t-1} = vp_{x+t-1} \cdot {}_tV_{x:\overline{n}}^1$$

```
1 Lemma res_term_rec : forall t x n : nat, 0 < t <= n -> x+t < \omega ->
2 \V_{(t-1) & x'1:n} + \P_{x'1:n} - \v * \q_{(x+t-1)} = \v * \p_{(x+t-1)} * \V_{t & x'1:n}.
```

また、Coq は極限も扱えるので、次のような連続払と分割払の関係式も証明可能である。

$$\lim_{m \rightarrow \infty} A_{x:\overline{n}}^{1(m)} = \bar{A}_{x:\overline{n}}^1$$

```
1 Lemma lim_ins_term_cont : forall x n : nat,
2 is_lim_seq (fun m:nat => \A^{m}_{x'1:n}) \Abar_{x'1:n}.
```

これらは Actuary パッケージにおいて形式的な証明を与えられているため、論理的には一切誤りのないことが保証されている。実際の証明やその他の形式化された補題等については、GitHub にアップロードしている Actuary パッケージを直接ご覧いただきたい。

2.3 Actuary パッケージの応用例

実際に Actuary パッケージを用いてどのように証明を行うのか、具体的な応用例を以下に示す。これらは Actuary パッケージのファイル Examples.v から抜粋している。イメージが湧くように証明スクリプトも提示するが、詳細な説明は省略する。

■生命年金現価が予定利率に関し広義単調減少であること 予定利率 i, i' に関する生命年金現価をそれぞれ $\ddot{a}_{x:\overline{n}|}$, $\ddot{a}'_{x:\overline{n}|}$ とする時, $i \leq i'$ ならば $\ddot{a}_{x:\overline{n}|} \geq \ddot{a}'_{x:\overline{n}|}$ が成り立つ。これは Actuary パッケージを使用して次のように証明することができる。

```

1 Lemma ann_due_decr_i : forall (i i' : R) (x n : nat), 0 < i -> 0 < i' -> x < \omega ->
2   i <= i' -> \a''[i']_{x:n} <= \a''[i]_{x:n}.
3 Proof.
4   move => i i' x n Hipos Hi'pos Hx Hlei'.
5   have Hvpos : 0 < \v[i] by apply /v_pos /Hipos.
6   have Hv'pos : 0 < \v[i'] by apply /v_pos /Hi'pos.
7   rewrite !ann_due_annual.
8   apply Rsum_le_compat => k /andP; case => /leP Hmk /ltP Hkn.
9   apply Rmult_le_compat_r; [by apply (p_nonneg _ l_fin) |].
10  case: (zerop k) => [Hk0 | Hkpos].
11  - rewrite Hk0 !Rpower_0 //; lra.
12  - case: (Rle_lt_or_eq_dec i i') => // [Hlt | Heq].
13    + rewrite /Rpower.
14      apply /Rlt_le /exp_increasing.
15      apply Rmult_lt_compat_l; [rewrite (_ : 0 = INR 0%N) //; apply lt_INR => // |].
16      apply ln_increasing => //.
17      rewrite /v_pres.
18      apply Rinv_1_lt_contravar; lra.
19      + rewrite Heq; lra.
20 Qed.

```

■将来法の責任準備金と過去法の責任準備金が一致すること Actuary パッケージでは責任準備金は将来法で定義されている。次の例は、養老保険の責任準備金について、過去法で計算しても結果が一致することを証明したものである。

```

1 Theorem res_eq_pros_retro : forall (t x n : nat), x+t < \omega -> t <= n -> 0 < n ->
2   \V_{t & x:n} = \P_{x:n} * \s''_{x:t} - \A_{x'1:t} / \A_{x:t'1}.
3 Proof.
4   move => t x n Hxt Htn Hn.
5   have Hx : x < \omega by apply lt_INR; move: Hxt; rewrite -plus_INR => /INR_lt; lia.
6   have Ha'' : \a''_{x : n} <> 0.
7   { apply /pos_neq0.
8     eapply Rlt_le_trans; [| apply (ann_due_lb _ i_pos _ l_fin) => //].
9     by apply Rinv_0_lt_compat. }
10  rewrite (_ : \s''_{x:t} = (\a''_{x:n} / \A_{x:t'1} - \a''_{(x+t):(n-t)})).
11  2:{ rewrite -(acc_due_plus_ann_due _ _ _ t) //}.

```



```

12      rewrite /Rdiv (Rmult_assoc _ \A_{x:t'1}) Rinv_r;
13      [| apply /pos_neq0 /(ins_pure_endow_pos _ _ l_fin) => //]; lra. }
14      rewrite Rmult_plus_distr_l.
15      rewrite {1}/prem_endow_life.
16      rewrite /Rdiv -Rmult_assoc (Rmult_assoc \A_{_:}) Rinv_l // Rmult_1_r.
17      rewrite /Rminus Rplus_assoc (Rplus_comm _ (-)) -Rplus_assoc
18      -/(Rminus (_*_)) (_*_)) -Rmult_minus_distr_r.
19      rewrite (_ : (\A_{x:n} - \A_{x'1:t}) */ \A_{x:t'1} = \A_{(x+t):(n-t)}).
20      2:{ rewrite (ins_endow_pure_endow _ _ _ t) //.
21      rewrite Rplus_comm Ropp_r_simpl_l Rmult_comm -Rmult_assoc Rinv_l;
22      [| by apply /pos_neq0 /(ins_pure_endow_pos _ _ l_fin)]; lra. }
23      rewrite /res_endow_life; lra.
24      Qed.

```

■Thiele の微分方程式 生命関数に連続微分可能性を仮定すれば, 例えば養老保険について Thiele の微分方程式

$$\frac{d}{dt} {}_t\bar{V}_{u:\overline{n}}^{(\infty)} = \bar{P}_{u:\overline{n}}^{(\infty)} - \mu_{u+t} + (\delta + \mu_{u+t}) \cdot {}_t\bar{V}_{u:\overline{n}}^{(\infty)}$$

が成立することも形式的に証明できる.

```

1 Theorem Thiele_ODE : forall t u n : R, u+t < \psi ->
2   is_derive (fun t:R => \Vbar^{p_infty}_{t & u:n}) t
3   (\Pbar^{p_infty}_{u:n} - \mu_{(u+t)} + (\delta + \mu_{(u+t)}) * \Vbar^{p_infty}_{t & u:n}).

```

なお, この証明スクリプトは約 90 行にわたり徒に紙面を消費するので割愛する.

3 今後の展望と課題

最後に, Coq や Actuary パッケージの将来的な応用可能性, および今後の課題について考察する.

3.1 アクチュアリー業務への応用可能性

今まで定理証明支援系の学術的な側面を中心に述べてきたが, 実務への応用も十分に考えられる.

3.1.1 保険数理に関する文書の校閲

定理証明支援系が証明の無謬性を保証する有効な手段である以上, 保険数理に関する文書に記載された数式の検証は Actuary パッケージの最も直接的な応用であると考えられる.

■Actuary パッケージの応用方法 Coq および Actuary パッケージでの検証が可能だと期待される対象としては, 例えば

1. 保険料及び責任準備金の算出方法書,
2. アクチュアリー資格試験問題

が考えられる.

算出方法書については、営業保険料や責任準備金の算式が商品仕様と整合していることを Coq で証明できれば、記載内容の正確性が格段に向上する。非常に簡単な例ではあるが、いわゆる α - β - γ 方式で計算された営業保険料の形式的検証を試みる。例えば

予定新契約費 新契約時にのみ、保険金額 1 に対し α ,

予定集金費 保険料払込みの都度、営業保険料 1 に対し β ,

予定維持費 保険料払込みの都度、毎年始に保険金額 1 に対し γ

の経費を織込んだ営業保険料の計算式が

$$P_{x:\overline{n}}^* = \frac{A_{x:\overline{n}} + \alpha + \gamma \ddot{a}_{x:\overline{n}}}{(1 - \beta) \ddot{a}_{x:\overline{n}}}$$

であると算出方法書に記載されていたとする。この算式を検証するため、営業保険料 $P_{x:\overline{n}}^*$ が収支相等の原則を満たすことを Coq で証明する。スクリプトの詳細は Actuary パッケージの Examples.v に格納されているが、以下はそこから主要部分を抜粋したものである。

```

1  (* initial expense *)
2  (* The expense  $\alpha$  per unit insurance amount is paid at the beginning of the first year. *)
3  Variable  $\alpha$ :R.
4  Hypothesis alpha_pos : 0 <  $\alpha$ .
5
6  (* collection expense *)
7  (* The expense  $\beta$  per unit premium is paid at the beginning of each year. *)
8  Variable  $\beta$ :R.
9  Hypothesis beta_in_0_1 : 0 <  $\beta$  < 1.
10
11 (* maintenance expense *)
12 (* The expense  $\gamma$  per unit insurance amount is paid at the beginning of each year. *)
13 Variable  $\gamma$ :R.
14 Hypothesis gamma_pos : 0 <  $\gamma$ .
15
16 (* annual gross premium of an endowment life insurance *)
17 Definition gross_prem_endow_life_annual (x n : nat) :=
18   ( $\backslash A_{\{x:n\}} + \alpha + \gamma * \backslash a''_{\{x:n\}}$ ) / ((1- $\beta$ ) *  $\backslash a''_{\{x:n\}}$ ).
19 Notation " $\backslash Pstar_{\{ x : n \}}$ " := (gross_prem_endow_life_annual x n) (at level 9, x at level 9).
20
21 (* Verify that the gross premium satisfies the equivalence principle. *)
22 Theorem gross_prem_endow_life_annual_correct : forall x n : nat, x <  $\backslash \omega$  -> 0 < n ->
23    $\backslash Pstar_{\{x:n\}} * \backslash a''_{\{x:n\}} = \backslash A_{\{x:n\}} + \alpha + (\backslash Pstar_{\{x:n\}} * \beta + \gamma) * \backslash a''_{\{x:n\}}$ .
24 Proof.
25   move => x n Hx Hn.
26   rewrite /gross_prem_endow_life_annual.
27   field; split; [| lra].
28   apply pos_neq0; eapply Rlt_le_trans;
29     [| apply (ann_due_lb _ i_pos _ l_fin)]; first apply Rinv_0_lt_compat; auto.
30 Qed.

```

最後の Theorem `gross_prem_endow_life_annual_correct` において、前述の営業保険料が収支相等の原則

$$P_{x:\overline{n}}^* \ddot{a}_{x:\overline{n}} = A_{x:\overline{n}} + \alpha + (P_{x:\overline{n}}^* \beta + \gamma) \ddot{a}_{x:\overline{n}}$$

を満たすことが形式的に証明されている。

なお、保険料算出原理には唯一の絶対的な方法があるわけではなく、各生命保険会社が十分性や公平性等を考慮して合理的な方法を選択することとされている。Coq を利用して営業保険料の算式を検証する場合は、算出方法書に記載された算式が、その保険会社によって採用された保険料算出原理に基づくことを形式的に証明すればよいであろう。ただし、保険料算出原理自体の妥当性は定理証明支援系の検証対象外である点には留意すべきである。アクチュアリー数学に限ったことではないが、定理証明支援系が担保するのは検証対象が形式的に記述された仕様を満たすことであり、その仕様自体の正当性ではない。つまり、保険料算出原理自体が妥当であるかどうかは引き続き人間が判断しなければならない、その原理を正確に形式仕様記述言語へ翻訳する作業も、やはり人間が行わなければならない。

続いて、Coq を用いたアクチュアリー資格試験問題の検証を試みる。具体的には、検証対象の問題文を仮定して解答を形式的に証明できれば、出題ミスを排除することが可能になるであろう。具体例として、生保数理 (2019 年度)[28] の問題 1.(3) を取上げる。

問題 1. 次の (1) ～ (6) について、各問の指示に従い、解答用紙の所定の欄にマークしなさい。

(中略)

(3) $A_x = 0.8499$ 、 $A_{x+1} = 0.8573$ 、予定利率 $i = 1.50\%$ のとき、 p_x の値に最も近いものは次のうちどれか。

- | | | | | |
|------------|------------|------------|------------|------------|
| (A) 0.9610 | (B) 0.9613 | (C) 0.9616 | (D) 0.9619 | (E) 0.9622 |
| (F) 0.9625 | (G) 0.9628 | (H) 0.9631 | (I) 0.9634 | (J) 0.9637 |

この問題の解答は $p_x = 0.962519$ により (F) となっている。これを `Actuary` パッケージで形式的に検証した実例を同パッケージの `Examples.v` に掲載している。以下はそこから結論部分を引用したものである。

```

1 Theorem Exam_2019_1_3 : forall x:nat, x+1 < \omega ->
2   \A_x = 0.8499 -> \A_(x+1) = 0.8573 -> \i = 0.015 -> 0.96235 < \p_x < 0.96265.
3 Proof.
4   move => x Hx HAx HAx1 Hi.
5   have Hpx : \p_x = (1 - (1+\i)*\A_x) / (1 - \A_(x+1)).
6   { move: (ins_whole_rec \i i_pos 1 l_fin x Hx).
7     rewrite (_ : \q_{0|1&x} = 1 - \p_x);
8     [| rewrite -{2}(p_q_1 1 1 x); [lra | apply /pos_neq0 /l_x_pos; lra]].
9     rewrite /v_pres => ->; field; lra. }
10  rewrite Hpx HAx HAx1 Hi; lra.
11 Qed.

```


一般に形式的証明を記述するにはかなりの労力を要するため、例えば生保数理でよく使用される式変形等を Coq で自動化できれば、実務への応用がより容易になるであろう。

■今後の課題 ただし、現状では前述のような形式的検証が可能な範囲は極めて限られていると言わざるを得ない。なぜならば、現時点の `Actuary` パッケージで形式化できているのは生保数理のごく一部であるからであ

る。実際、実務でよく利用される多重脱退残存表や計算基数等は **Actuary** パッケージに実装されていない。さらに、今後の形式化にあたって課題になり得る点として次のようなものが考えられる。

近似計算の取扱い 生保数理ではその歴史的な経緯もあり種々の近似計算が行われる。しかし、Coq ではたとえどんなに誤差が小さくても異なる実数であれば等号でつなぐことができない。この課題への対処方法は近似の種類によって変わってくると思われる。例えば Taylor 展開を用いた近似であれば極限を用いて誤差項を評価することが考えられる。また、場合によっては不等式を用いることで近似を表現することも可能かもしれない。

生命関数の微分可能性 現時点の **Actuary** パッケージでは、生命関数の微分を行う際、実数全体での連続微分可能性を仮定している。その場合、例えば直線で表される生命関数でさえも分析の対象外となってしまう。



もちろん、連続微分可能性の条件を緩められれば問題はない。しかし、**Actuary** パッケージで使用している Coquelicot ライブラリでは、形式化された定理の多くで関数に強い微分可能性が要求されている。そのため、生命関数に仮定する条件を緩め過ぎると既存のライブラリを利用できなくなる懸念がある。解決策の一つとして、**Actuary** パッケージでの応用に耐えられるように Coquelicot ライブラリの定理を一般化することが考えられる。ただし、一般に数学の形式化に要する工数を事前に見積もるのは困難であることが知られており、筆者自身、この解決策を実行に移す場合の労力や所要時間を見通すことができない。

Coq における解析学の形式化 現状、解析学における広義積分は Coq で十分に形式化されていないため、**Actuary** パッケージでも広義積分を取扱っていない。そのため、例えば死力が一定であるような生命関数は **Actuary** パッケージの適用対象外である。広義積分の形式化はそう簡単ではなく、**Actuary** パッケージのためだけに臨時で準備するのは現実的でない。解決策としては、Coq による解析学の形式化の進展を待つこと、あるいは別の定理証明支援系を利用することが考えられる。前者に関しては、現在 Reynald Affeldt を中心に MathComp-Analysis²² というライブラリの開発が進められており、そこでは Coq によって Lebesgue 積分論が形式化されているため、このライブラリが完成すれば **Actuary** パッケージで扱える範囲も広がることが期待される。さらに、Stieltjes 積分を多用する損保数理の形式化も実現できる可能性がある。ただし、MathComp-Analysis は Coquelicot と互換性がないため、MathComp-Analysis を採用する場合は **Actuary** パッケージを抜本的に修正する必要があると思われる。もう一つの解決策としては、解析学の形式化が進んでいる定理証明支援系で別途生保数理を形式化することが考えられる。例えば第 1.1.2 節で紹介した定理証明支援系 Isabelle は解析学の形式化がかなり進んでいるため有力な候補と考えられる。実際、筆者は現在 Isabelle による生保数理の形式化を試みている。

3.1.2 実務で使用されるプログラムの検証

第 1.2.2 節でも言及した通り、Coq はプログラムの形式的な検証に利用できる。アクチュアリー業務においては種々の計算を行う上でさまざまなソフトウェアやプログラムが用いられるため、それらの検査に **Actuary** パッケージが応用できる可能性がある。現時点では実際の応用まで研究が進んでいないので、関連する研究を

紹介することで今後の発展の方向性を探ることとしたい。

■Coq から検証済みのプログラムを抽出する方法 Coq にはもともとプログラム抽出という機能が備わっており、Coq で記述された関数等を所定の関数型言語のソースコードへ変換することが可能である。従って、

1. Coq で所望の関数を記述する、
2. その関数が目的の仕様を満たすことを Coq で証明する、
3. プログラム抽出を実行する、

という手順を踏むことで、バグのないソースコードが得られる。ただし、論文執筆時点で正式にサポートされている出力先の言語は OCaml, Haskell, Scheme の 3 種類である。この背景には、そもそも Coq の仕様記述言語が Gallina という関数型言語であるため、抽出先も関数型言語である方が実装が容易である、といった理由があるものと思われる。これら以外のプログラミング言語への抽出は Coq の標準機能には備わっていない。そのため、Coq から他の言語、例えば Scala [11, 18], C [31], Java [34], Ruby [27], Python [36] 等へのプログラム抽出に関する研究が行われている。

■既存のプログラムを Coq に読み込んで検証する方法 逆に、既存のプログラムを Coq に読み込んで検証を行う方法もある。その場合、

1. 既存のソースコードからモデルを記述する、
2. そのモデルが目的の仕様を満たすことを Coq で証明する、

という手順を踏むことになる。実例としては、Java で書かれた既存のメールサーバのプログラムを Coq で検証したケーススタディが Reynald Affeldt 等により報告されている [1]。また、Coq 以外の定理証明支援系や自動定理証明機も用いてソースコードを検証する手段として Why3²³ という有名なソフトウェアもある。これを用いる場合は、検証対象のプログラムを Why3 上で証明できるように WhyML という中間言語へ翻訳する作業が必要になる。

なお、C については強力な検証プラットフォームとして VST (Verified Software Toolchain)²⁴ が開発されている。これを利用すると、C のソースコードから形式的検証を経たアセンブリコードを生成することができる。VST は以下の点で優れたツールであると考えられる。

CompCert の利用 C には形式的検証を経た CompCert²⁵ というコンパイラがある。そもそもソースコードが正しくても、コンパイラ自体にバグがあれば誤ったアセンブリコードが生成されてしまう。CompCert については、ソースコードとコンパイル後のアセンブリコードの等価性が Coq により証明されている。VST は CompCert との親和性が高く、VST で検証した C のソースコードを CompCert でコンパイルすれば一切のバグが除かれたアセンブリコードが得られる。

自動翻訳の利用 CompCert に含まれている `clightgen` というツールを用いると、C のソースコードを Coq 上の抽象構文に翻訳することができる。つまり、ソースコードをモデル化する作業が自動化されるため、プログラム検証の労力を削減できる。

VST については開発者 Andrew W. Appel 等により詳しい入門書 [3] が出版されている。また、VST を用いて Newton 法の近似精度を形式的に証明したケーススタディも報告されている [2]。

■**アクチュアリー業務への応用方法** 以上のように, Coq を始めとする定理証明支援系や自動定理証明機を用いてプログラムの検証を行う研究はさまざまな角度から行われている. 実際にどのようなアプローチを採用するかは, 実務でどのプログラミング言語を使用しているか, ソースコードやアセンブリコードにどの程度の信頼性を求めるか, 確保されている予算やリソースはどの程度か, 等により異なってくると思われる. もしバグを完全に排除するのであれば, VST のように

1. ソースコードが所望の仕様を満たしていること,
2. 使用するコンパイラにバグがないこと

を形式的に証明することが求められよう. しかし, この目標を完全に達成するには膨大な工数がかかると見込まれる上, 場合によってはソフトウェア開発の専門家の協力も必要になると思われる. 実務上どこまで妥協できるのか, どのように工夫すれば工数が抑えられるのかといった考察は今後の重要な研究テーマとなり得る.

3.2 数学の形式化の未来

定理証明支援や自動定理証明の技術が進歩し, 現代数学の形式化がさらに進んだ未来ではどのようなことが起こるだろうか. 将来, 人工知能が人間の数学者を凌駕する日が来るだろうか. これは深い問いであり, それをテーマとして一冊の本ができるほどである [33].

コンピュータによる証明が今後どこまで進歩するかは分からないが, 少なくとも現在, 我々がその目覚ましい発展の渦中にいることは間違いない. 筆者も数学を専攻した者としてアクチュアリーの立場からこの研究に少しでも貢献していきたい.

付録

定理証明支援系はその有用性にもかかわらず日本における知名度が低いように感じている. 本論文を執筆した動機の一つは, 数学の予備知識を持つアクチュアリーの方々へ Coq の存在を知ってもらうことであった.

Coq の定番の入門書としては Yves Bertot, Pierre Castéran による “Interactive Theorem Proving and Program Development” [6] がある. 計算機科学者向けのテキストとしては Benjamin C. Pierce 等による “Software Foundations” というシリーズがあり, 以下の Web サイトから閲覧可能である.

<https://softwarefoundations.cis.upenn.edu/>

このシリーズの一部は日本語にも翻訳されている.

<https://www.chiguri.info/sfja/>

また, 日本語で書かれた初学者向けの記事として池淵未来のチュートリアルがある.

<https://www.iiijlab.net/activities/programming-coq/>

Actuary パッケージで利用している SSReflect/MathComp については Yves Bertot 等によるオンラインの入門書 “Mathematical Components” [7] が無料で閲覧可能である. 日本語の入門書としては, 萩原学, Reynald Affeldt による “Coq/SSReflect/MathComp による定理証明” [16] が 2018 年に出版された.

Coq のコミュニティは全世界に広がっており, Coq のメーリングリスト `coq-club`²⁶ や Zulip のチャット

ルーム²⁷では毎日世界中の Coq ユーザの間で活発なやり取りがなされている。本論文が Coq コミュニティに新たな参加者が加わるきっかけとなれば誠に喜ばしい。また、アクチュアリー数学の形式化に興味をお持ちの方は筆者までご連絡いただきたい。

謝辞

アクチュアリー数学の形式化にあたり、Coq を始めとする定理証明支援系の研究状況につき、産業技術総合研究所の Reynald Affeldt 氏より貴重な助言をいただきました。心より御礼申し上げます。

また、2021 年 11 月 5 日に開催された 2021 年度日本アクチュアリー会年次大会では論文発表および質疑応答を通じて本研究への新たな視点を得ることができました。大会委員会を始め年次大会を運営いただいた方々に改めて御礼申し上げます。

そして、2021 年 11 月 21 日-22 日に開催された The 17th Theorem Proving and Provers meeting (TPP 2021)²⁸ では、本研究を定理証明支援系の専門家へ発表する機会を得るとともに、今後の発展に関し重要な示唆をいただきました。本研究集会の幹事である名古屋大学の才川隆文氏、北見工業大学の松田一徳氏にこの場をお借りして感謝の意を表します。

さらに、スウェーデン王立工科大学の Karl Palmskog 氏には Actuary パッケージの OPAM リポジトリへの登録、およびインストールに必要な環境の整備を行っていただいた上、本研究をソフトウェア検証に応用する方法についてご教示いただきました。浅学の筆者に丁寧な説明をいただいたことを深く感謝申し上げます。

注釈

¹ 本論文の内容は全て筆者個人の見解であり、所属する組織としての見解を示すものではない。また、筆者および所属組織は、本研究に起因するいかなる損害に対しても責任を負わない。

²<https://jscoq.github.io/>

³<https://coq.inria.fr/distrib/current/stdlib/>

⁴<https://coq.inria.fr/packages.html>

⁵<https://math-comp.github.io/>

⁶<http://coquelicot.saclay.inria.fr/>

⁷<https://www.isa-afp.org/>

⁸<https://leanprover.github.io/live/latest/>

⁹https://leanprover-community.github.io/mathlib_docs/

¹⁰ 数学においては通常、“命題”とは証明された論理式を表すが、ここでは単なる主張という意味で用い、その真偽にかかわらず“命題”と呼ぶ。

¹¹ 説明の都合上一部簡略化している。

¹² 二重否定の除去は直観論理では認められていないが、ここでは通常の数学で使用する古典論理を前提として議論を進める。

¹³ “Robbins 代数がブール代数になる”という予想。20 世紀前半に Herbert Robbins によって提示された。

¹⁴ “無限の空間において同半径の球の敷き詰めを行った時、最密充填は面心立方格子である”という予想。Johannes Kepler により 1611 年に予想された。

¹⁵<http://www.cs.ru.nl/F.Wiedijk/100/index.html>

¹⁶ ここでは“三段論法”を意味する英単語“syllogism”を用いた。

¹⁷ 型 `Prop` は“命題”を意味する英単語“proposition”に由来する。なお、通常の数学では命題を全称記号で量化することはないと思われるが、Coq は高階述語論理を扱えるのでこのような記述が可能である。

¹⁸ ここで `H.PQR` というのは仮定の名前であり、原則として自由に名づけることができる。

¹⁹ GitHub とはソフトウェア開発のために利用されるインターネット上のプラットフォームであり、定理証明支援系 Coq の開発も GitHub で行われている。

²⁰ Coq のファイルは拡張子を `.v` とする。

²¹ 一般に、関数 f の x における値を f_x のように表す。

²² <https://github.com/math-comp/analysis>

²³ <http://why3.lri.fr/>

²⁴ <https://vst.cs.princeton.edu/>

²⁵ <https://compcert.org/>

²⁶ coq-club@inria.fr

²⁷ <https://coq.zulipchat.com>

²⁸ <https://t6s.github.io/tpp2021/>

参考文献

- [1] Reynald Affeldt, Naoki Kobayashi, and Akinori Yonezawa. Verification of concurrent programs using the Coq proof assistant: A case study. *IPSJ Digital Courier*, Vol. 1, pp. 117–127, 2005.
- [2] Andrew W. Appel and Yves Bertot. C floating-point proofs layered with VST and Flocq. *Journal of Formalized Reasoning*, Vol. 13, No. 1, pp. 1—16, 2020.
- [3] Andrew W. Appel, et al. *Program Logics for Certified Compilers*. Cambridge University Press, 2014.
- [4] Kenneth Appel and Wolfgang Haken. Every planar map is four colorable. *Bulletin of the American Mathematical Society*, Vol. 82, No. 5, pp. 711–712, 1976.
- [5] Jeremy Avigad, Johannes Hölzl, and Luke Serafin. A formally verified proof of the central limit theorem. *Journal of Automated Reasoning*, Vol. 59, pp. 389–423, 2017.
- [6] Yves Bertot and Pierre Castéran. *Interactive Theorem Proving and Program Development: Coq’Art: The Calculus of Inductive Constructions*. Springer-Verlag Berlin Heidelberg New York, 2004.
- [7] Yves Bertot, Georges Gonthier, Assia Mahboubi, and Enrico Tassi. Mathematical Components. <https://zenodo.org/record/4457887#.YCjx12NUtpQ>, 2021.
- [8] Kevin Buzzard. Proving theorems with computers. *Notices of the American Mathematical Society*, Vol. 67, No. 11, pp. 1791–1799, 2020.
- [9] Thierry Coquand and Gérard Huet. The calculus of constructions. Technical Report RR-0530, INRIA, 1986.
- [10] Leonardo de Moura, Soonho Kong, Jeremy Avigad, Floris van Doorn, and Jakob von Raumer. The Lean theorem prover (system description). In Amy P. Felty and Aart Middeldorp, editors, *Automated Deduction - CADE-25*, pp. 378–388. Springer International Publishing, 2015.
- [11] Youssef El Bakouny and Dani Mezher. Scallina: Translating verified programs from Coq to Scala. In Sukyoung Ryu, editor, *Programming Languages and Systems*, pp. 131–145. Springer International

- Publishing, 2018.
- [12] 二見隆. 生命保険数学 (’92改訂版) 上巻. 生命保険文化研究所, 1992.
 - [13] 二見隆. 生命保険数学 (’92改訂版) 下巻. 生命保険文化研究所, 1992.
 - [14] Georges Gonthier. A computer-checked proof of the four colour theorem. <http://www2.tcs.ifi.lmu.de/~abel/lehre/WS07-08/CAFR/4colproof.pdf>, 2005.
 - [15] Sébastien Gouëzel and Vladimir Shchur. A corrected quantitative version of the Morse lemma. *Journal of Functional Analysis*, Vol. 277, No. 4, pp. 1258–1268, 2019.
 - [16] 萩原学, アフェルトレナルド. Coq/SSReflect/MathComp による定理証明: フリーソフトではじめる数学の形式化. 森北出版, 2018.
 - [17] Thomas C. Hales. Formal proof. *Notices of the American Mathematical Society*, Vol. 55, No. 11, pp. 1370–1380, 2008.
 - [18] 逸見港, 田辺良則, 今井宜洋, 萩谷昌己. 検証済みのコードによる Coq から Scala へのコード抽出. 日本ソフトウェア科学会第 31 回大会講演論文集, 2014.
 - [19] 情報処理推進機構. セキュリティ機能と保証レベル. <https://www.ipa.go.jp/security/jisec/forusers/abouteal.html>, 2020.
 - [20] Chester Wallace Jordan, Jr. *Society of Actuaries’ Textbook on Life Contingencies*. The Society of Actuaries, second edition, 1967.
 - [21] Cezary Kaliszyk and Josef Urban. Learning-assisted automated reasoning with Flyspeck. *Journal of Automated Reasoning*, Vol. 53, pp. 173–213, 2014.
 - [22] Vladislav Kargin. Life contingency models I. https://www2.math.binghamton.edu/lib/exe/fetch.php/people/kargin/publications/actuarial_1_ln_master.pdf, 2017.
 - [23] Gerwin Klein, et al. seL4: Formal verification of an OS kernel. In *Proceedings of the ACM SIGOPS 22nd Symposium on Operating Systems Principles*, SOSP ’09, pp. 207–220, 2009.
 - [24] Johnny Li and Andrew Ng. *SOA Exam MLC Study Manual*. ACTEX Learning, fall 2017 edition, 2017. Also available as <https://www.actexamdriver.com/trials/Actex%20MLC%20Fall%202017%20Print%20FT%20sample.pdf>.
 - [25] The mathlib Community. The Lean mathematical library. In *Proceedings of the 9th ACM SIGPLAN International Conference on Certified Programs and Proofs*, No. 15 in CPP 2020, pp. 367–381. Association for Computing Machinery, 2020.
 - [26] William Mccune. Solution of the Robbins problem. *Journal of Automated Reasoning*, Vol. 19, pp. 263–276, 1997.
 - [27] Hiroki Mizuno. coq-ruby. <https://github.com/mzp/coq-ruby>, 2010.
 - [28] 日本アクチュアリー会. 資格試験過去問題集 (2019 年度): 生保数理. <http://www.actuaries.jp/lib/collection/books/2019/2019B.pdf>, 2019.
 - [29] Lawrence C. Paulson. Natural deduction as higher-order resolution. *Journal of Logic Programming*, Vol. 3, pp. 237–258, 1986.
 - [30] Eric V. Slud. Actuarial mathematics and life-table statistics. <https://www.math.umd.edu/~slud/s470/BookChaps/01Book.pdf>, 2001.
 - [31] Akira Tanaka. Coq to C translation with partial evaluation. In *Proceedings of the 2021 ACM SIGPLAN Workshop on Partial Evaluation and Program Manipulation*, PEPM 2021, pp. 14–31.

- Association for Computing Machinery, 2021.
- [32] The Coq Development Team. The Coq reference manual: Release 8.15.2. <https://coq.inria.fr/distrib/current/refman/>, May 2022.
 - [33] 照井一成. コンピュータは数学者になれるのか?: 数学基礎論から証明とプログラムの理論へ. 青土社, 2015.
 - [34] 渡辺優樹. 定理証明支援系言語 Coq からの Java プログラム抽出. http://onct.oita-ct.ac.jp/seigyo/nishimura_hp/pdf/2017watanabe.pdf, 2018.
 - [35] Freek Wiedijk. Formal proof: Getting started. *Notices of the American Mathematical Society*, Vol. 55, No. 11, pp. 1408–1414, 2008.
 - [36] 山本晃治, 宗像一樹. 証明駆動開発の現実的な開発プロジェクトへの適用に向けて. ソフトウェアエンジニアリングシンポジウム 2016 論文集, Vol. 2016, pp. 104–111, 2016.